Subject: Re: Extracting the resultant angles from the Hough transform
Posted by txominhermos on Thu, 03 Nov 2005 11:53:32 GMT
View Forum Message <> Reply to Message

THE ANSWER TO THIS QUESTION HAS BEEN GIVEN BY THE RSI TECH SUPPORT

The THETA output variable can be used to find the angle for a line.
Here is a short example demonstrating how this is done:

```
;************************************************************ ****************
pro test

; Choose two points in the range
; 1 < x < 199
; 1 < y < 199
p0 = [40.,20]
p1 = [120,180]

nx = 200d
ny = 200d

array = fltarr(nx,ny)
array[[p0[0],p1[0]],[p0[1],p1[1]]] = 1b

; Display the data:
window, 0, xs = nx, ys = ny, $
xpos = 0, ypos = 0, $
title = 'original data'
tvscl, array

; HOUGH input parameters:
dx = 1.2
dy = 1.0
xmin = 0.0
ymin = 0.0

r = hough(array, rho = rho, theta = theta, $
dx = dx, dy = dy, xmin = xmin, ymin = ymin)

sz = size(r, /dim)

; Locate the line
maxcount = where(r eq max(r))

; Use the output variable THETA to get the line's angle:
```

```
radangles = theta[maxcount mod sz[0]] - !pi/2
print, 'Angle(s) : '
print, radangles
print

end
;************************************************************ ****************
```

Reply to this post